Type Normalization in R$\omega\mu$

ALEX HUBERS, The University of Iowa, USA

1 INTRODUCTION

We describe the normalization-by-evaluation (NBE) of types in R$\omega\mu$. Types are normalized modulo β- and η-equivalence—that is, to $\beta\eta$-long forms. Because the type system of R$\omega\mu$ is a strict extension of System F$\omega\mu$, type level computation for arrow kinds is isomorphic to reduction of arrow types in the STLC. Novel to this report are the reductions of Π, Σ, and label bound terms.

2 SYNTAX OF KINDS

Our formalization of R$\omega\mu$ types is *intrinsic*, meaning we define the syntax of *typing* and *kinding judgments*, foregoing any description of untyped syntax. The syntax of types is indexed by kinding environments and kinds, defined below.

```
data Kind : Set where
  ★     : Kind
  L     : Kind
  _‘→_ : Kind → Kind → Kind
  R[_] : Kind → Kind

infixr 5 _‘→_
```

The kind system of R$\omega\mu$ defines $\star$ as the type of types; L as the type of labels; $(\rightarrow)$ as the type of type operators; and $R[\kappa]$ as the type of *rows* containing types at kind κ. As shorthand, we write $R^n[\kappa]$ to denote n repeated applications of R to the type κ–e.g., $R^3[\kappa]$ is shorthand for $R[R[R[\kappa]]]$.

The syntax of kinding environments is given below. Kinding environments are isomorphic to lists of kinds.

```
data KEnv : Set where
  ϵ : KEnv
  _„_ : KEnv → Kind → KEnv
```

Let the metavariables Δ and κ range over kinding environments and kinds, respectively. Correspondingly, we define *generalized variables* in Agda at these names.

```
private
  variable
    Δ Δ₁ Δ₂ Δ₃ : KEnv
    κ κ₁ κ₂ : Kind
```

The syntax of intrinsically well-scoped De-Bruijn type variables is given below. We say that the kind variable x is indexed by kinding environment Δ and kind κ to specify that x has kind κ in kinding environment Δ.

Author's address: Alex Hubers, Department of Computer Science, The University of Iowa, 14 MacLean Hall, Iowa City, Iowa, USA, alexander-hubers@uiowa.edu.

```
data KVar : KEnv → Kind → Set where
  Z : KVar (Δ „ κ) κ
  S : KVar Δ κ₁ → KVar (Δ „ κ₂) κ₁
```

3 SYNTAX OF TYPES

Rωμ is a qualified type system with predicates of the form $\rho_1 \lesssim \rho_2$ and $\rho_1 \cdot \rho_2 \sim \rho_3$ for row-kinded types ρ_1, ρ_2, and ρ_3. Because predicates occur in types and types occur in predicates, the syntax of well-kinded types and well-kinded predicates are mutually recursive. The syntax for each is given below. we describe (in this order) the syntactic components belonging to System Fωμ, qualified type systems, and system Rω.

```
data Pred (Δ : KEnv) : Kind → Set
data Type Δ : Kind → Set
data Type Δ where

  ‘ :
    (α : KVar Δ κ) →
    Type Δ κ

  ‘λ :
    (τ : Type (Δ „ κ₁) κ₂) →
    Type Δ (κ₁ ‘→ κ₂)

  _·_ :
    (τ₁ : Type Δ (κ₁ ‘→ κ₂)) →
    (τ₂ : Type Δ κ₁) →
    Type Δ κ₂

  _‘→_ :
    (τ₁ : Type Δ ★) →
    (τ₂ : Type Δ ★) →
    Type Δ ★

  ‘∀ :
    (τ : Type (Δ „ κ) ★) →
    Type Δ ★

  μ :
    (F : Type Δ (★ ‘→ ★)) →
    Type Δ ★
```

The first three constructors are analogous to the terms of the STLC. the constructor `→ classifies term functions; the constructor `∀ classifies type-in-term quantification; and the constructor μ classifies recursive terms. Note that μ could be further generalized to kind κ `→ ★; however, we found that kind ★ `→ ★ was sufficient for our needs while simplifying both presentation and mechanization.

The syntax of qualified types is given below.

```
_⇒_ :
  (π : Pred Δ R[ κ₁ ]) → (τ : Type Δ ★) →
  Type Δ ★
```

The type $\pi \Rightarrow \tau$ states that τ is *qualified* by the predicate π—that is, the type variables bound in τ are restricted in instantiation to just those that satisfy the predicate π. This is completely analogous to identical syntax used in Haskell to introduce typeclass qualification. Predicates are defined below (after the presentation of type syntax).

We now describe the syntax exclusive to Rωμ, beginning with label kind introduction and elimination. Labels are first-class entities in Rωμ, and may be represented by both constants and variables.

```
lab :
  (l : Label) →
  Type Δ L

⌊_⌋ :
  (τ : Type Δ L) →
  Type Δ ★
```

Label constants in Rωμ are constructed from the type `Label`; in our mechanization, `Label` is a type synonym for `String`, but one could choose any other candidate with decidable equality. Types at label kind `L` may be cast to *label singletons* by the ⌊_⌋ constructor. This makes labels first-class entities: for example, as the type `⌊ lab "l" ⌋` has kind ★, it can be inhabited by a term.

Types at row kind are constructed by one of the following three constructors.

```
ϵ :
  Type Δ R[ κ ]

_▹_ :
  (l : Type Δ L) → (τ : Type Δ κ) →
  Type Δ R[ κ ]

_<$>_ :
  (f : Type Δ (κ₁ ‘→ κ₂)) → (τ : Type Δ R[ κ₁ ]) →
  Type Δ R[ κ₂ ]
```

Rows in Rωμ are either the empty row ϵ, a labeled row `(l ▹ τ)`, or a row mapping `f <$> τ`. The row mapping `f <$> (l ▹ τ)` describes the lifting of the function `f` over row `(l ▹ τ)`, which we will define to equal `(l ▹ f τ)` in the case where the right hand applicand is a labeled row. We will show that rows in Rome (that is, types at row kind) reduce to either the empty row ϵ or a labeled row `(l ▹ τ)` after normalization. There are two important consequences of this canonicity: firstly, we treat row mapping `_<$>_` as having latent computation to perform (there are no normal types with form `f <$> τ` except when τ is a neutral variable). The second consequence is that we do not permit the formation of rows with more than one label-type association. Such rows are instead formed as type variables with predicates specifying the shape of the row.

Rows in Rωμ are eliminated by the Π and Σ constructors.

```
Π :
  Type Δ (R[ κ ] `→ κ)

Σ :
  Type Δ (R[ κ ] `→ κ)
```

Given a type ρ at row kind, $\Pi\rho$ constructs a record with label-type associations from ρ and $\Sigma\rho$ constructs a variant that has label and type from ρ. We choose to represent Π and Σ as type constants at kind `(R[ κ ] `→ κ)`; we will show that many applications of Π and Σ induce type reductions, and hence it is convenient to group such reductions with type application.

The syntax of predicates is given below. The predicate $\rho_1 \lesssim \rho_2$ states that label-to-type mappings in ρ_1 are a subset of those in ρ_2; the predicate $\rho_1 \cdot \rho_2 \sim \rho_3$ states that the combination of mappings in ρ_1 and ρ_2 equals ρ_3.

```
data Pred Δ where
  _·_~_ :
    (ρ₁ ρ₂ ρ₃ : Type Δ R[ κ ]) →
    Pred Δ R[ κ ]

  _≲_ :
    (ρ₁ ρ₂ : Type Δ R[ κ ]) →
    Pred Δ R[ κ ]
```

Hubers and Morris [2023] implicitly define two type-level row lifting operators, *left mapping* <$> and *right mapping* <?>, but the latter is superfluous. We appeal to the kinds of these operators for their intuition: left mapping `f <$> ρ` lifts a function at arrow kind `f : κ₁ → κ₂` into a function at kind `R[ κ₁ ] → R[ κ₂ ]` and then applies it to `ρ : R[ κ₂ ]`. We may define right mapping (named *flap* and written <?>, after similar Haskell operators) of row function $f : R[\kappa_1 \to \kappa_2]$ over type $\tau : \kappa_1$ using left mapping under the following identity:

$$f \mathbin{<?>} \tau = (\lambda\, g.\, g\, \tau) \mathbin{<\$>} f$$

which we encode in Agda as follows:

```
flap : Type Δ (R[ κ₁ `→ κ₂ ] `→ κ₁ `→ R[ κ₂ ])
flap = `λ (`λ ((` Z) · (` (S Z)))) <$> (` (S Z))))

_<?>_ : Type Δ (R[ κ₁ `→ κ₂ ]) → Type Δ κ₁ → Type Δ R[ κ₂ ]
f <?> a = flap · f · a
```

(We choose to define `_<?>_` as the application of `flap` to inputs `f` and `a` so that we needn't pollute the definition with weakenings of its arguments.)

3.1 Type renaming

We closely follow Wadler et al. [2022] and Chapman et al. [2019] in defining a *type renaming* as a function from type variables in one kinding environment to type variables in another. This is the *parallel renaming and substitution* approach for which weakening and single variable substitution are special cases. The code we establish now will be mimicked again for both normal types and for

terms; many names are reused, and so we find it helpful to index duplicate names by a suffix. The suffix $_k$ specifies that this definition describes the Type syntax.

```
Renaming_k : KEnv → KEnv → Set
Renaming_k Δ₁ Δ₂ = ∀ {κ} → KVar Δ₁ κ → KVar Δ₂ κ
```

We will let the metavariable ρ range over both renamings and types at row kind.

Lifting can be thought of as the weakening of a renaming, and permits renamings to be pushed under binders.

```
lift_k : Renaming_k Δ₁ Δ₂ → Renaming_k (Δ₁ „ κ) (Δ₂ „ κ)
lift_k ρ Z = Z
lift_k ρ (S x) = S (ρ x)
```

We define renaming as a function that translates a kinding derivation in kinding environment Δ_1 to environment Δ_2 provided a renaming from Δ_1 to Δ_2. The definition proceeds by induction on the input kinding derivation. In the variable case, we use ρ to rename variable x. In the `λ and `∀ cases, we must lift the renaming ρ over the type variable introduced by these binders. The rest of the cases are effectively just congruence over the type structure.

```
ren_k : Renaming_k Δ₁ Δ₂ → Type Δ₁ κ → Type Δ₂ κ
renPred_k : Renaming_k Δ₁ Δ₂ → Pred Δ₁ R[ κ ] → Pred Δ₂ R[ κ ]

ren_k ρ (` x)        = ` (ρ x)
ren_k ρ (`λ τ)       = `λ (ren_k (lift_k ρ) τ)
ren_k ρ (π ⇒ τ)      = renPred_k ρ π ⇒ ren_k ρ τ
ren_k ρ (`∀ τ)       = `∀ (ren_k (lift_k ρ) τ)
ren_k ρ ϵ            = ϵ
ren_k ρ (τ₁ · τ₂)    = (ren_k ρ τ₁) · (ren_k ρ τ₂)
ren_k ρ (τ₁ `→ τ₂)   = (ren_k ρ τ₁) `→ (ren_k ρ τ₂)
ren_k ρ (μ F)        = μ (ren_k ρ F)
ren_k ρ Π            = Π
ren_k ρ Σ            = Σ
ren_k ρ (lab x)      = lab x
ren_k ρ (l ▹ τ)      = ren_k ρ l ▹ ren_k ρ τ
ren_k ρ ⌊ ℓ ⌋        = ⌊ (ren_k ρ ℓ) ⌋
ren_k ρ (f <$> m)    = ren_k ρ f <$> ren_k ρ m
```

As Type and Pred are mutually inductive, we must define `renPred`$_k$ as mutually recursive to `ren`$_k$. Its definition is completely unsuprising.

```
renPred_k ρ (ρ₁ · ρ₂ ~ ρ₃) = ren_k ρ ρ₁ · ren_k ρ ρ₂ ~ ren_k ρ ρ₃
renPred_k ρ (ρ₁ ≲ ρ₂) = (ren_k ρ ρ₁) ≲ (ren_k ρ ρ₂)
```

Finally, weakening is a special case of renaming.

```
weaken_k : Type Δ κ₂ → Type (Δ „ κ₁) κ₂
weaken_k = ren_k S
```

3.2 Type substitution

We wish to give both a declarative and algorithmic treatment of type equivalence. For the latter, we will normalize types to normal forms, meaning types are equivalent iff their normal forms are definitionally equal. For the former, we must define β-substitution syntactically so that we can express β-equivalence of types declaratively. In our development, β-reduction is a special case of substitution.

We define a substitution as a function mapping type variables in context Δ_1 to types in context Δ_2.

```
Substitution_k : KEnv → KEnv → Set
Substitution_k Δ₁ Δ₂ = ∀ {κ} → KVar Δ₁ κ → Type Δ₂ κ
```

Substitutions must be lifted over binders, just as is done for renamings.

```
lifts_k : Substitution_k Δ₁ Δ₂ → Substitution_k(Δ₁ „ κ) (Δ₂ „ κ)
lifts_k σ Z = ‘ Z
lifts_k σ (S x) = weaken_k (σ x)
```

Substitution is defined inductively over types in a similar fashion to renaming. Note that this is *simultaneous* substitution and renaming—The variable case translates type variable `x` to the type $\sigma\ \tau$, for which the substitution σ also performs a renaming from environment Δ_1 to Δ_2. The rest of the cases (as with renaming) are either congruences over the type structure or congruences plus lifting of the substitution. Again, substitution over predicates is defined mutually recursively.

```
sub_k : Substitution_k Δ₁ Δ₂ → Type Δ₁ κ → Type Δ₂ κ
subPred_k : Substitution_k Δ₁ Δ₂ → Pred Δ₁ κ → Pred Δ₂ κ
sub_k σ ϵ = ϵ
sub_k σ (‘ x) = σ x
sub_k σ (‘λ τ) = ‘λ (sub_k (lifts_k σ) τ)
sub_k σ (τ₁ · τ₂) = (sub_k σ τ₁) · (sub_k σ τ₂)
sub_k σ (τ₁ ‘⟶ τ₂) = (sub_k σ τ₁) ‘⟶ (sub_k σ τ₂)
sub_k σ (π ⟹ τ) = subPred_k σ π ⟹ sub_k σ τ
sub_k σ (‘∀ τ) = ‘∀ (sub_k (lifts_k σ) τ)
sub_k σ (μ F) = μ (sub_k σ F)
sub_k σ (Π) = Π
sub_k σ Σ = Σ
sub_k σ (lab x) = lab x
sub_k σ (l ▹ τ) = sub_k σ l ▹ sub_k σ τ
sub_k σ ⌊ ℓ ⌋ = ⌊ (sub_k σ ℓ) ⌋
sub_k σ (f <$> a) = sub_k σ f <$> sub_k σ a

subPred_k σ (ρ₁ · ρ₂ ~ ρ₃) = sub_k σ ρ₁ · sub_k σ ρ₂ ~ sub_k σ ρ₃
subPred_k σ (ρ₁ ≲ ρ₂) = (sub_k σ ρ₁) ≲ (sub_k σ ρ₂)
```

We define the extension of a substitution σ by the type τ functionally. If we had chosen to represent a `Substitution_k` as a list, extension would be done by the `cons` constructor. In a De-Bruijn representation, the most recently appended variable is zero—hence an extension here maps the zero variable to τ in the `Z` case and maps each variable `(S x)` to its value in σ at predecessor `x`.

```
extend_k : Substitution_k Δ₁ Δ₂ → (τ : Type Δ₂ κ) → Substitution_k (Δ₁ „ κ) Δ₂
extend_k σ τ Z = τ
extend_k σ τ (S x) = σ x
```

Finally, β-substitution is simply a special case of substitution. Note that the constructor `` ` `` has type `KVar Δ κ → Type Δ κ`, making it a substitution. It is in fact an identity substitution, which fixes the meaning of its type variables, hence it is the substitution we choose to extend when defining β-substitution.

```
_β_k[_] : Type (Δ „ κ₁) κ₂ → Type Δ κ₁ → Type Δ κ₂
τ₁ β_k[ τ₂ ] = sub_k (extend_k ‘ τ₂) τ₁
```

4 TYPE EQUIVALENCE

We define type and predicate equivalence mutually recursively. You may think of type equivalence also as a sort of small-step relation on types, as we include rules to equate β-equivalent and η-equivalent types, as well as a number of computational steps a row kinded type may take.

```
infix 0 _≡t_
infix 0 _≡p_
data _≡p_ : Pred Δ R[ κ ] → Pred Δ R[ κ ] → Set
data _≡t_ : Type Δ κ → Type Δ κ → Set
```

Unless otherwise quantified, let the metavariable `l` range over types with label kind, let π range over predicates, and let τ and υ range over types:

```
private
  variable
    l l₁ l₂ l₃ : Type Δ L
    ρ₁ ρ₂ ρ₃ : Type Δ R[ κ ]
    π₁ π₂ : Pred Δ R[ κ ]
    τ τ₁ τ₂ τ₃ υ υ₁ υ₂ υ₃ : Type Δ κ
```

The rules for predicate equivalence are uninteresting: two predicates are considered equivalent when their component types are equivalent.

```
data _≡p_ where
  _eq-≲_ :
    τ₁ ≡t υ₁ → τ₂ ≡t υ₂ →
    τ₁ ≲ τ₂ ≡p υ₁ ≲ υ₂

  _eq-·_~_ :
    τ₁ ≡t υ₁ → τ₂ ≡t υ₂ → τ₃ ≡t υ₃ →
    τ₁ · τ₂ ~ τ₃ ≡p υ₁ · υ₂ ~ υ₃
```

The first three rules for type equivalence state that it is an equivalence relation.

```
data _≡t_ where
  eq-refl :
    τ ≡t τ
```

```
  eq-sym :
    τ₁ ≡t τ₂ →
    τ₂ ≡t τ₁

  eq-trans :
    τ₁ ≡t τ₂ → τ₂ ≡t τ₃ →
    τ₁ ≡t τ₃
```

Type equivalence is congruent over the total structure of types, including λ-bindings (hence you may view type normalization as being *call-by-value*). We omit the other eight congruence rules.

```
eq-λ : ∀ {τ υ : Type (Δ „ κ₁) κ₂} →
  τ ≡t υ →
  `λ τ ≡t `λ υ
```

We have one η-equivalence rule. It is henceforth useful to view the following rules as directed left-to-right, as normal forms are produced on the right-hand side.

```
eq-η : ∀ {f : Type Δ (κ₁ `→ κ₂)} →
  f ≡t `λ (weaken_k f · (` Z))
```

The rules that remain as *computational*—these are precisely the rules we would use to define small-step reduction of types. We begin with the β-equivalence rule, which states that lambda abstractions applied to arguments are equivalent to their beta reduction.

```
eq-β : ∀ {τ₁ : Type (Δ „ κ₁) κ₂} {τ₂ : Type Δ κ₁} →
  ((`λ τ₁) · τ₂) ≡t (τ₁ β_k[ τ₂ ])
```

The next two rules specify the computational behavior of mapping over rows. Rule (eq-<$>$\epsilon$) states that mapping over the empty row ϵ should yield the empty row; rule eq-▹$ states that mapping over a labeled row should push the left applicand into the body of the row.

```
eq-<$>ϵ : {F : Type Δ (κ₁ `→ κ₂)} →
  (F <$> ϵ) ≡t ϵ

eq-▹$ : ∀ {l} {τ : Type Δ κ₁} {F : Type Δ (κ₁ `→ κ₂)} →
  (F <$> (l ▹ τ)) ≡t (l ▹ (F · τ))
```

We wish to establish that normal forms of types at row kind are either the empty row ϵ or labeled rows. This is, of course, not the case for types in general. For example, the type Π · (l ▹ τ) has row kind when τ has row kind R[κ]. In this case, rule eq-Π▹ pushes the Π over the label so that a canonical form is restored.

```
eq-Π▹ : ∀ {l} {τ : Type Δ R[ κ ]} →
  Π · (l ▹ τ) ≡t (l ▹ (Π · τ))
```

The application of Π and Σ to a type τ at nested-row kind is in fact just the mapping of Π and Σ over τ:

```
eq-Π : ∀ {τ : Type Δ R[ R[ κ ] ]} →
  Π · τ ≡t Π <$> τ
```

Likewise to rows, we wish to show that normal forms of types at arrow kind are canonically λ-bound. However, the type $\Pi \cdot (\mathtt{l} \triangleright \mathtt{`}\lambda\ \tau)$ has arrow kind! Rule eq-Π$\lambda$ pushes the $\lambda$ outwards in order to restore canonicity and so that application of $\Pi \cdot (\mathtt{l} \triangleright \mathtt{`}\lambda\ \tau)$ to an applicand is simply β-reduction.

```
eq-Πλ : ∀ {l} {τ : Type (Δ ,, κ₁) κ₂} →
  Π · (l ▹ ‘λ τ) ≡t ‘λ (Π · (weaken_k l ▹ τ))
```

Finally, in many cases (such as record concatenation and variant branching) it is necessary to reassociate the application $(\Pi\ \rho)\ \tau$ inward so that Π (or Σ) are the outermost syntax. We observe the following reassociation identity:

```
eq-Π-assoc : ∀ {ρ : Type Δ (R[ κ₁ ‘⟶ κ₂ ])} {τ : Type Δ κ₁} →
  (Π · ρ) · τ ≡t Π · (ρ <?> τ)
```

The definition of `_≡t_` concludes by repeating the last four rules, replacing each Π with Σ. As a final aside, it might be thought that we could have rid ourselves of the syntax for mapping by elaborating types at kind $\mathtt{R[}\ \kappa_1 \longrightarrow \kappa_2\ \mathtt{]}$. For example, the type $(\mathtt{l} \triangleright \lambda\ \mathtt{x} : \kappa_1.\ \tau)$ could perhaps have its λ binding pushed outside to yield $\lambda\ \mathtt{x} : \kappa_1.\ (\mathtt{l} \triangleright \tau)$. However, this would not be kind-preserving (the latter has kind $\kappa_1 \longrightarrow \mathtt{R[}\ \kappa_2\ \mathtt{]}$), and therefore such a translation would induce a normalization that does not preserve kinds. We believe it would be possible but complicated to consider a kind-changing translation.

5 NORMAL TYPES

As is common in other *normalization by evaluation* approaches, we separate *neutral types* from *normal types*. These two definitions are defined mutually inductively with the data type for normal predicates:

```
data NormalType (Δ : KEnv) : Kind → Set
data NormalPred (Δ : KEnv) : Kind → Set
data NeutralType Δ : Kind → Set
```

A type is neutral if it is (respectively) (i) a variable, (ii) the application of a variable to an argument, or (iii) the mapping of a normal function type over a neutral row type. Intuitively, neutral forms are forms for which computation is "stuck" waiting on a variable to be substituted for a canonical form. Note that this third neutral form (row mapping) is novel to our development, and, in comparison to application, inverts the normal/neutral expectation of its arguments. It captures the stuck nature of a type such as $(\mathtt{l} \triangleright \lambda\ \mathtt{x}.\ \mathtt{M})\ \mathtt{<\$>}\ \rho$—that is, we are unable to map a function over a type variable.

```
data NeutralType Δ where
  ‘_ :
    (α : KVar Δ κ) →
    NeutralType Δ κ

  _·_ :
    (f : NeutralType Δ (κ₁ ‘⟶ κ)) →
    (τ : NormalType Δ κ₁) →
    NeutralType Δ κ

  _<$>_ :
```

```
    (F : NormalType Δ (κ₁ ‘⟶ κ₂)) → (τ : NeutralType Δ R[ κ₁ ]) →
    NeutralType Δ (R[ κ₂ ])
```

A predicate is normal if its component types are each normal.

```
data NormalPred Δ where
  _·_~_ :
    (ρ₁ ρ₂ ρ₃ : NormalType Δ R[ κ ]) →
    NormalPred Δ R[ κ ]

  _≲_ :
    (ρ₁ ρ₂ : NormalType Δ R[ κ ]) →
    NormalPred Δ R[ κ ]
```

Because we consider the normalization of types modulo η-equivalence, we wish to restrict our normal types to η-long form. This can be done by restricting the construction of normal-neutral types to just ground kind. This also ensures a canonical form for arrow-kinded normal types, as neutral types at arrow-kind cannot be promoted to normal types. We define a `Ground` predicate on types that maps all non-arrow kinds to the unit type $\top$ and maps the arrow kind to $\bot$. (In other words, `Ground` κ is trivially inhabitable so long as $\kappa \neq \kappa_1 \to \kappa_2$.)

```
Ground : Kind → Set
Ground ★ = ⊤
Ground L = ⊤
Ground (κ ‘⟶ κ₁) = ⊥
Ground R[ κ ] = ⊤
```

It is easy to show that this predicate is decidable.

```
ground? : ∀ κ → Dec (Ground κ)
ground? ★ = yes tt
ground? L = yes tt
ground? (_ ‘⟶ _) = no (λ ())
ground? R[ _ ] = yes tt
```

Now we may restrict the `ne` constructor to promoting just neutral types at ground kind by adding the (implicit) requirement that `ne` only be used when `Ground` κ is satisfied. To make this evidence easy to populate when κ is known, we employ a well-known proof-by-reflection trick (see Wadler et al. [2022]) and require evidence of the form `True (ground? κ)`.

```
data NormalType Δ where
  ne :
    (x : NeutralType Δ κ) → {ground : True (ground? κ)} →
    NormalType Δ κ
```

Likewise, to ensure canonical forms of rows, we restrict Π and Σ to formation at kind $\star$ and L. The constructors for record types are given below.

```
Π :
  (ρ : NormalType Δ R[ ⋆ ]) →
  NormalType Δ ⋆

ΠL :
  (ρ : NormalType Δ R[ L ]) →
  NormalType Δ L
```

The rest of the `NormalType` syntax is identical to the `Type` syntax with the exception that we remove the ` constructor for variables and Π and Σ constructors at arbitrary kind. We choose not to omit this syntax, as our proofs of canonicity follow from knowing the totality of `NormalType` constructors.

```
- Fω
`λ :
  (τ : NormalType (Δ „ κ₁) κ₂) →
  NormalType Δ (κ₁ `→ κ₂)

_`→_ :
  (τ₁ τ₂ : NormalType Δ ⋆) →
  NormalType Δ ⋆

`∀ :
  {κ : Kind} → (τ : NormalType (Δ „ κ) ⋆) →
  NormalType Δ ⋆

μ :
  (F : NormalType Δ (⋆ `→ ⋆)) →
  NormalType Δ ⋆

- Qualified types
_⇒_ :
  (π : NormalPred Δ R[ κ₁ ]) → (τ : NormalType Δ ⋆) →
  NormalType Δ ⋆

- Rω
ϵ :
  NormalType Δ R[ κ ]


_▹_ :
  (l : NormalType Δ L) →
  (τ : NormalType Δ κ) →
  NormalType Δ R[ κ ]

lab :
  (l : Label) →
  NormalType Δ L

⌊_⌋ :
  (l : NormalType Δ L) →
```

```
  NormalType Δ ★
Σ :
  (ρ : NormalType Δ R[ ★ ]) →
  NormalType Δ ★

ΣL :
  (ρ : NormalType Δ R[ L ]) →
  NormalType Δ L
```

5.1 Renaming

We define renaming over `NormalTypes` in the same fashion as defined over `Types`. Note that we use the suffix $_k$NF now to denote functions which operate on `NormalType` syntax. Definitions are unsurprising and omitted.

```
renₖNE      : Renamingₖ Δ₁ Δ₂ → NeutralType Δ₁ κ → NeutralType Δ₂ κ
renₖNF      : Renamingₖ Δ₁ Δ₂ → NormalType Δ₁ κ → NormalType Δ₂ κ
weakenₖNF : NormalType Δ κ₂ → NormalType (Δ „ κ₁) κ₂
weakenₖNE : NeutralType Δ κ₂ → NeutralType (Δ „ κ₁) κ₂
```

5.2 Properties of normal types

We use Agda to confirm the desired canonicity properties. First, we wish for arrow kinds to be canonically formed by λ-abstractions. This can be shown easily by induction on arrow-kinded `f`.

```
arrow-canonicity : (f : NormalType Δ (κ₁ ‘→ κ₂)) → ∃[ τ ] (f ≡ ‘λ τ)
arrow-canonicity (‘λ f) = f , refl
```

Second, we wish for types at row kind to be canonically either (i) a labeled type `(l ▹ τ)`, (ii) a neutral type, or (iii) the empty row ϵ. The `row-canonicity` lemma below states precisely this. Note that we permit row-kinded types to be neutral because we do not η-expand arrow-kinded rows. Recall our discussion above that such an expansion would not be kind-preserving. This means arrow-kinded rows must be permitted to be canonically neutral.

```
row-canonicity : (ρ : NormalType Δ R[ κ ]) →
  ∃[ l ] (Σ[ τ ∈ NormalType Δ κ ] ((ρ ≡ (l ▹ τ)))) or
  Σ[ τ ∈ NeutralType Δ R[ κ ] ] (ρ ≡ ne τ) or
  ρ ≡ ϵ
row-canonicity (l ▹ τ) = left (l , τ , refl)
row-canonicity (ne τ) = right (left (τ , refl))
row-canonicity ϵ = right (right refl)
```

5.3 Type embeddings

We establish an embedding back from normal types to types below. The embedding is written ⇑ because its type is converse to our definition of normalization, written ⇓. We will show in later sections precisely that ⇑ is right-inverse to ⇓.

```
⇑ : NormalType Δ κ → Type Δ κ
⇑NE : NeutralType Δ κ → Type Δ κ
⇑Pred : NormalPred Δ R[ κ ] → Pred Δ R[ κ ]
```

Much of the embedding is defined by using like-for-like constructors and recursing on the subdata.

```
⇑NE (‘ x) = ‘ x
⇑NE (τ₁ · τ₂) = (⇑NE τ₁) · (⇑ τ₂)
⇑NE (F <$> τ) = (⇑ F) <$> (⇑NE τ)

⇑Pred (ρ₁ · ρ₂ ~ ρ₃) = (⇑ ρ₁) · (⇑ ρ₂) ~ (⇑ ρ₃)
⇑Pred (ρ₁ ≲ ρ₂) = (⇑ ρ₁) ≲ (⇑ ρ₂)

⇑ ϵ = ϵ
⇑ (ne x) = ⇑NE x
⇑ (l ▹ τ) = (⇑ l) ▹ (⇑ τ)
⇑ (‘λ τ) = ‘λ (⇑ τ)
⇑ (τ₁ ‘⟶ τ₂) = ⇑ τ₁ ‘⟶ ⇑ τ₂
⇑ (‘∀ τ) = ‘∀ (⇑ τ)
⇑ (μ τ) = μ (⇑ τ)
⇑ (lab l) = lab l
⇑ ⌊ τ ⌋ = ⌊ ⇑ τ ⌋
⇑ (π ⟹ τ) = (⇑Pred π) ⟹ (⇑ τ)
```

An exception is made for record and variant constructors, which we must reconstruct as applications:

```
⇑ (Π x) = Π · ⇑ x
⇑ (ΠL x) = Π · ⇑ x
⇑ (Σ x) = Σ · ⇑ x
⇑ (ΣL x) = Σ · ⇑ x
```

6 SEMANTIC TYPES

We next define `SemType Δ κ`, the semantic interpretation of types. `SemTypes` are defined by induction on the kind κ and mutually-recursively with `KripkeFunctions`, the interpretation of type functions.

```
SemType : KEnv → Kind → Set
KripkeFunction : KEnv → Kind → Kind → Set
```

Type functions are interpreted as Kripke function spaces because they must permit arbitrary and intermediate renaming. That is, they are functions at "any world."

```
KripkeFunction Δ₁ κ₁ κ₂ = (∀ {Δ₂} → Renaming_k Δ₁ Δ₂ → SemType Δ₂ κ₁ → SemType Δ₂ κ₂)
SemType Δ₁ (κ₁ ‘⟶ κ₂) = KripkeFunction Δ₁ κ₁ κ₂
```

We interpret ★ and L kinded types as their normal forms.

```
SemType Δ ★ = NormalType Δ ★
SemType Δ L = NormalType Δ L
```

We interpret rows as either `nothing` (the empty row), `just (left x)` for neutral `x`, or `just (right (l , τ))` for normal `l` and τ. These cases correspond precisely to the three canonical forms of types with row kind.

```
SemType Δ R[ κ ] = Maybe
  ((NeutralType Δ R[ κ ]) or
  (NormalType Δ L × SemType Δ κ))
```

6.1 Renaming & substitution

Renaming is defined over semantic types in an obvious fashion. Definitions are omitted except in the functional case.

```
renSem : Renaming_k Δ₁ Δ₂ → SemType Δ₁ κ → SemType Δ₂ κ
weakenSem : SemType Δ κ₁ → SemType (Δ „ κ₂) κ₁
```

Because R$\omega\mu$ functions are interpreted into Kripke function spaces, renaming of arrow-kinded types is simply composition by the function's renaming.

```
renKripke : Renaming_k Δ₁ Δ₂ → KripkeFunction Δ₁ κ₁ κ₂ → KripkeFunction Δ₂ κ₁ κ₂
renKripke {Δ₁} ρ F {Δ₂} = λ ρ' → F (ρ' ∘ ρ)

renSem {κ = κ ‘→ κ₁} ρ F = renKripke ρ F
```

6.2 Normalization by evaluation

Our *normalization by evaluation* proceeds in a standard fashion. We will define `reflect`, which maps neutral types to semantic types, and `reify`, which maps semantic types to normal types. We then write an evaluator that takes a `Type` into the semantic domain. During this process, function applications (and other forms of computation) are reduced. We finally reify the semantic type back to a normal form.

Reflection and reification are defined mutually recursively. We define the type synonym `reifyKripke` the reification of types at arrow kind, for repeated use later.

```
reflect : ∀ {κ} → NeutralType Δ κ → SemType Δ κ
reify : ∀ {κ} → SemType Δ κ → NormalType Δ κ
reifyKripke : KripkeFunction Δ κ₁ κ₂ → NormalType Δ (κ₁ ‘→ κ₂)
```

Reflection of neutral types at ground kind leaves the type undisturbed.

```
reflect {κ = ★} τ       = ne τ
reflect {κ = L} τ       = ne τ
reflect {κ = R[ κ ]} τ = just (left τ)
```

Reflection of neutral types at arrow kind must be η-expanded into a Kripke function. Note here that is necessary to reify the input `v` back to a normal type.

```
reflect {κ = κ₁ ‘→ κ₂} τ = λ ρ v → reflect (ren_kNE ρ τ · reify v)
```

Reification similarly leaves ground types undisturbed. Semantic types at ★ and label kind are already in normal form; semantic types at row kind must be translated from their semantic constructors to their `NormalType` constructors.

```
reify {κ = ★} τ = τ
reify {κ = L} τ = τ
reify {κ = R[ κ ]} (just (left x)) = ne x
```

```
reify {κ = R[ κ ]} (just (right (l , τ))) = l ▹ (reify τ)
reify {κ = R[ κ ]} nothing = ϵ
```

Semantic functions must be reified from Agda functions back into `NormalType` syntax. This is done by reifying the application of semantic function `F` to the reflection of the η-expanded variable `` ` Z ``.

```
reify {κ = κ₁ `→ κ₂} F = reifyKripke F
reifyKripke {κ₁ = κ₁} F = `λ (reify (F S (reflect {κ = κ₁} (` Z))))
```

Observe that neutral types can be forced into η-long form simply by composing reification and reflection. This will prove helpful later, as the neutral type former `ne` has the same type except restricted to ground kind, but we will need to be able to promote from neutral to normal type at *all* kinds.

```
η-norm : NeutralType Δ κ → NormalType Δ κ
η-norm = reify ∘ reflect
```

Towards writing an evaluator, we define a semantic environment as a function mapping type variables to semantic types.

```
Env : KEnv → KEnv → Set
Env Δ₁ Δ₂ = ∀ {κ} → KVar Δ₁ κ → SemType Δ₂ κ
```

Environment extension and lifting can be written in a straightforward manner.

```
extende : (η : Env Δ₁ Δ₂) → (V : SemType Δ₂ κ) → Env (Δ₁ ,, κ) Δ₂
lifte : Env Δ₁ Δ₂ → Env (Δ₁ ,, κ) (Δ₂ ,, κ)
```

The identity environment now maps type variables to semantic types. Unlike in Chapman et al. [2019], this environment can no longer be truly said to be an identity: type variables are de facto put into η-long form during reflection. However this change is mandatory for normalization, so we cannot define an environment that does not.

```
idEnv : Env Δ Δ
idEnv = reflect ∘ `
```

6.3 Helping evaluation

In aid of writing an evaluator, we found it helpful to develop *semantic* notions of the syntax introduced by R$\omega\mu$. For example, we define a type synonym for application, which is simply Agda application within the identity renaming.

```
_·V_ : SemType Δ (κ₁ `→ κ₂) → SemType Δ κ₁ → SemType Δ κ₂
F ·V V = F id V
```

We can further define the constructors of the three canonical forms of row-kinded types:

```
_▹V_ : SemType Δ L → SemType Δ κ → SemType Δ R[ κ ]
_▹V_ {κ = κ} ℓ τ = just (right (ℓ , τ))

ne-R : NeutralType Δ R[ κ ] → SemType Δ R[ κ ]
ne-R = just ∘ left
```

```
ϵV : SemType Δ R[ κ ]
ϵV = nothing
```

The definition of semantic row mapping varies by the shape of the row `V` over which we are lifting. If `V` is neutral, so too must the mapping of `F` over `!V!` be neutral. Hence we reify `F` to normal form and leave its mapping in neutral form. If `V` is a labeled row `(l ▹ τ)`, we push the application of `F` over τ. Finally, if `V` is the empty row, its mapping is empty.

```
_<$>V_ : SemType Δ (κ₁ ‘⟶ κ₂) ⟶ SemType Δ R[ κ₁ ] ⟶ SemType Δ R[ κ₂ ]
_<$>V_ {κ₁ = κ₁} {κ₂} F (just (left x)) = ne-R (reifyKripke F <$> x)
_<$>V_ {κ₁ = κ₁} {κ₂} F (just (right (l , τ))) = (l ▹V (F ·V τ))
_<$>V_ {κ₁ = κ₁} {κ₂} F nothing = ϵV
```

Although the flap operator `_<?>_` is expressible as a special case of row mapping, we nevertheless find it a useful abstraction to express as a semantic function. It is defined below in terms of semantic row mapping; we find it likewise helpful to give a type synonym `apply` to the left hand side of this equation.

```
apply : SemType Δ κ₁ ⟶ SemType Δ ((κ₁ ‘⟶ κ₂) ‘⟶ κ₂)
apply a = λ ρ F ⟶ F ·V (renSem ρ a)

infixr 0 _<?>V_
_<?>V_ : SemType Δ R[ κ₁ ‘⟶ κ₂ ] ⟶ SemType Δ κ₁ ⟶ SemType Δ R[ κ₂ ]
f <?>V a = apply a <$>V f
```

Much of the latent computation in R$\omega\mu$ occurs under an outermost Π and Σ syntax. To this end, we chose to represent Π and Σ as arrow-kinded type-constants—meaning they will evaluate into Agda functions. This provides an opportunity to concisely abstract their reduction logic. We define a semantic combinator for the Π type constant below. The first two equations state that record types at ⋆ and label kind may be formed provided normal bodies; The third equation pushes the λ-binding of F outside of the record type; the fourth equation states that application *is* mapping at nested row kind.

```
ΠV : SemType Δ R[ κ ] ⟶ SemType Δ κ
ΠV {κ = ⋆} x = Π (reify x)
ΠV {κ = L} x = ΠL (reify x)
ΠV {κ = κ₁ ‘⟶ κ₂} F = λ ρ v ⟶ ΠV (renSem ρ F <?>V v)
ΠV {κ = R[ κ ]} x = (λ ρ v ⟶ ΠV v) <$>V x
```

We can turn the semantic helper `ΠV` into a true Kripke function easily:

```
Π-Kripke : KripkeFunction Δ R[ κ ] κ
Π-Kripke = λ ρ v ⟶ ΠV v
```

We omit the definitions of `ΣV` and `Σ-Kripke`, as they are identical modulo the use of Π constants.

6.4 Evaluation

We now write an evaluator that translates Types to semantic types; that is, translating syntactic forms to the semantic domain. A normalizer composes reification with evaluation. One can see this in the definition of `evalPred`, the predicate normalizer. (Predicates must be fully normalized as they do not have a semantic image.)

```
eval : Type Δ₁ κ → Env Δ₁ Δ₂ → SemType Δ₂ κ
evalPred : Pred Δ₁ R[ κ ] → Env Δ₁ Δ₂ → NormalPred Δ₂ R[ κ ]

evalPred (ρ₁ · ρ₂ ~ ρ₃) η = reify (eval ρ₁ η) · reify (eval ρ₂ η) ~ reify (eval ρ₃ η)
evalPred (ρ₁ ≲ ρ₂) η = reify (eval ρ₁ η) ≲ reify (eval ρ₂ η)
```

Evaluation is defined by induction over the type structure. The first three cases have types which may occur at any kind. The variable case simply uses the environment to perform a lookup; application defers to our semantic combinator _·V_; and evaluation of arrow types is defined recursively.

```
eval {κ = κ} (` x) η = η x
eval {κ = κ} (τ₁ · τ₂) η = (eval τ₁ η) ·V (eval τ₂ η)
eval {κ = κ} (τ₁ `→ τ₂) η = (eval τ₁ η) `→ (eval τ₂ η)
```

The next four cases are for types that only occur at kind ★. The qualified type and label singleton cases proceed by recursion over the type structure. For `∀-bound types, we must lift the environment $\eta$ appropriately. In the $\mu$ case, $\tau$ has kind $\star \to \star$ and so its evaluation must be reified back to `NormalType`.

```
eval {κ = ★} (π ⇒ τ) η = evalPred π η ⇒ eval τ η
eval {κ = ★} ⌊ τ ⌋ η = ⌊ eval τ η ⌋
eval {κ = ★} (`∀ τ) η = `∀ (eval τ (lifte η))
eval {κ = ★} (μ τ) η = μ (reify (eval τ η))
```

There is only one type with exclusively label kind. Its definition is unsurprising (it houses only a `String` label).

```
eval {κ = L} (lab l) η = lab l
```

We evaluate λ-bound functions by evaluating their bodies in environments extended by the meaning their input v. Note that we are building a Kripke function and so ρ is a renaming from Δ_1 to Δ_2 and v is an input of type `SemType` Δ_2 κ_1.

```
eval {κ = κ₁ `→ κ₂} (`λ τ) η = λ ρ v → eval τ (extende (λ {κ} v' → renSem {κ = κ} ρ (η v')) v)
```

Lastly, we define evaluation over the row-kinded constants and operators. As Π and Σ are represented as type constants in the Type syntax, they translate directly to the Kripke functions we defined for Π and Σ as semantic helpers. Likewise, the row mapping and labeled-row cases are interpreted immediately and desirably by their semantic helpers.

```
eval {κ = R[ κ ] `→ κ} Π η = Π-Kripke
eval {κ = R[ κ ] `→ κ} Σ η = Σ-Kripke
eval {κ = R[ κ ]} (f <$> a) η = (eval f η) <$>V (eval a η)
```

```
eval {κ = _} (l ▹ τ) η = (eval l η) ▹V (eval τ η)
eval ϵ η = ϵV
```

Finally, we define a normalizer as the reification of evaluation.

```
⇓ : ∀ {Δ} → Type Δ κ → NormalType Δ κ
⇓ τ = reify (eval τ idEnv)

⇓NE : ∀ {Δ} → NeutralType Δ κ → NormalType Δ κ
⇓NE τ = reify (eval (⇑NE τ) idEnv)
```

7 CORRECTNESS

We desire a normalization algorithm to remove the need for explicit type conversion proofs in terms: two types are equal iff they reduce to the same normal form, and so a normalization algorithm effectively gives a decision procedure for type equivalence. It next falls upon us to verify that this normalization algorithm indeed respects our syntactic account of type equivalence. How we do so is fairly routine to other normalization-by-evaluation efforts. We first show that the algorithm is complete with respect to syntactic type equivalence:

```
completeness : ∀ {τ₁ τ₂ : Type Δ κ} → τ₁ ≡t τ₂ → ⇓ τ₁ ≡ ⇓ τ₂
```

Completeness here states that equivalent types normalize to the same types. Conversely, soundness states that every type is equivalent to its normalization. (In particular, every type is equivalent to the normalization of its embedding.)

```
soundness : ∀ {Δ₁ κ} → (τ : Type Δ₁ κ) → τ ≡t ⇑ (⇓ τ)
```

A final (but no less crucial) property we will show is *stability*: that every normal type is equal to the normalization of its embedding.

```
stability : ∀ (τ : NormalType Δ κ) → ⇓ (⇑ τ) ≡ τ
```

It is desirable that a normalization algorithm adheres to this property, as it states effectively that there is "no more work" to be done by re-normalization. Indeed, both idempotency and surjectivity are implied.

```
idempotency : ∀ (τ : Type Δ κ) → (⇑ (⇓ (⇑ (⇓ τ)))) ≡ ⇑ (⇓ τ)
idempotency τ rewrite stability (⇓ τ) = refl

surjectivity : ∀ (τ : NormalType Δ κ) → ∃[ υ ] (⇓ υ ≡ τ)
surjectivity τ = ( ⇑ τ , stability τ )
```

7.1 A logical relation

We will prove completeness using a logical relation on semantic types. We would like to be able to equate semantic types, but they prove to be "too large": in particular, our definition of Kripke functions permit functions which may not respect composition of renaming. The solution is to reason about semantic types modulo a partial equivalence relation (PER) that both respects renamings (which we call *uniformity*) and also equates functions extensionally. We write $\tau_1 \approx \tau_2$ to denote that the semantic types τ_1 and τ_2 are equivalent modulo this relation. For clarity, we give names to the two properties (*uniformity* and *point equality*) we desire related types to hold, and define them mutually recursively.

```
_≈_ : SemType Δ κ → SemType Δ κ → Set
PointEqual-≈ : ∀ {Δ₁} {κ₁} {κ₂} (F G : KripkeFunction Δ₁ κ₁ κ₂) → Set
Uniform : ∀ {Δ} {κ₁} {κ₂} → KripkeFunction Δ κ₁ κ₂ → Set
```

We define `_≈_` recursively over the kind of its equated types. In the first two cases, τ_1 and τ_2 are normal types, which we equate propositionally. In the third case, we assert that Kripke functions `F` and `G` are uniform and point-equal to one another. Uniformity asserts a certain commutativity of renaming: you may either rename the result of applying `F` to $\mathtt{V}_1$, or you may rename `F` before applying it to a renamed input. point equality on Kripke functions `F` and `G` asserts that `F` and `G` take related inputs to related outputs. The latter property is what one should expect of a logical relation; the former property can be attributed to Chapman et al. [2019], who in turn attribute Allais et al. [2013].

```
Uniform {Δ₁} {κ₁} {κ₂} F =
  ∀ {Δ₂ Δ₃} (ρ₁ : Renaming_k Δ₁ Δ₂) (ρ₂ : Renaming_k Δ₂ Δ₃) (V₁ V₂ : SemType Δ₂ κ₁) →
  V₁ ≈ V₂ → (renSem ρ₂ (F ρ₁ V₁)) ≈ (renKripke ρ₁ F ρ₂ (renSem ρ₂ V₂))

PointEqual-≈ {Δ₁} {κ₁} {κ₂} F G =
  ∀ {Δ₂} (ρ : Renaming_k Δ₁ Δ₂) {V₁ V₂ : SemType Δ₂ κ₁} →
  V₁ ≈ V₂ → F ρ V₁ ≈ G ρ V₂

_≈_ {κ = ★} τ₁ τ₂ = τ₁ ≡ τ₂
_≈_ {κ = L} τ₁ τ₂ = τ₁ ≡ τ₂
_≈_ {Δ₁} {κ = κ₁ ‘→ κ₂} F G =
  Uniform F × Uniform G × PointEqual-≈ {Δ₁} F G
```

The last six cases are over row kinded semantic types. The first case states that neutral rows must be propositionally equal; the second states that two rows of the form $(\mathtt{l}_1 \triangleright \tau_1)$ and $(\mathtt{l}_2 \triangleright \tau_2)$ are related iff their labels are equal and their types are related. The third case states that the empty row is related to itself (which is always true). All other cases are nonsensical, and so are set to $\bot$.

```
_≈_ {κ = R[ κ ]} (just (left x)) (just (left y)) = x ≡ y
_≈_ {κ = R[ κ ]} (just (right (l₁ , τ₁))) (just (right (l₂ , τ₂))) = l₁ ≡ l₂ × τ₁ ≈ τ₂
_≈_ {κ = R[ κ ]} nothing nothing           = ⊤
_≈_ {κ = R[ κ ]} (just _) (just _)         = ⊥
_≈_ {κ = R[ κ ]} (just _) nothing          = ⊥
_≈_ {κ = R[ κ ]} nothing (just _)          = ⊥
```

7.1.1 Properties of the completeness relation. The completeness relation forms a *partial equivalence relation* (PER). As uniformity is a unary property, it follows quickly that `_≈_` cannot be reflexive, but a limited form of reflexivity does hold: provided that `V` is related to *some* other `V'`, it relates to itself. The other properties (symmetry and transitivity) are simple enough to show. We introduce two helpers, `refl-≈`$_l$ and `refl-≈`$_r$ to describe left and right reflexive projections.

```
refl-≈_l : ∀ {V₁ V₂ : SemType Δ κ} → V₁ ≈ V₂ → V₁ ≈ V₁
refl-≈_r : ∀ {V₁ V₂ : SemType Δ κ} → V₁ ≈ V₂ → V₂ ≈ V₂
sym-≈ : ∀ {τ₁ τ₂ : SemType Δ κ} → τ₁ ≈ τ₂ → τ₂ ≈ τ₁
trans-≈ : ∀ {τ₁ τ₂ τ₃ : SemType Δ κ} → τ₁ ≈ τ₂ → τ₂ ≈ τ₃ → τ₁ ≈ τ₃
```

we commonly invoke two main lemmas. `reflect-≈` states reflects propositional equality to semantic relatability, and `reify-≈` reifies related semantic types propositional equality. We make great use of the latter lemma, which states intuitively that related types should have the same reifications.

```
reflect-≈ : ∀ {τ₁ τ₂ : NeutralType Δ κ} → τ₁ ≡ τ₂ → reflect τ₁ ≈ reflect τ₂
reify-≈   : ∀ {τ₁ τ₂ : SemType Δ κ} → τ₁ ≈ τ₂ → reify τ₁ ≡ reify τ₂
```

TODO: congruence and commutativity files.

7.2 The fundamental theorem & completeness

We would like to show that all equivalent, well-kinded types inhabit the relation. Completeness follows shortly thereafter. The fundamental theorem for completeness (`fundC`) states that equivalent types evaluate to related types under related environments.

```
- fundC : ∀ {τ₁ τ₂ : Type Δ₁ κ} {η₁ η₂ : Env Δ₁ Δ₂} →
-        Env-≈ η₁ η₂ → τ₁ ≡t τ₂ → eval τ₁ η₁ ≈ eval τ₂ η₂
- fundC-pred : ∀ {π₁ π₂ : Pred Δ₁ R[ κ ]} {η₁ η₂ : Env Δ₁ Δ₂} →
-             Env-≈ η₁ η₂ → π₁ ≡p π₂ → evalPred π₁ η₁ ≡ evalPred π₂ η₂
```

8 SOUNDNESS

8.1 A logical relation

8.2 The fundamental theorem & soundness

9 STABILITY

10 REMARK

10.1 Comparison to Chapman et al. [2019]

Our mechanization has closely resembled that of Chapman et al. [2019]. Our definition of semantic types, however, has differed, as our normalization is with respect to both β- and η-equivalence, whereas Chapman et al's is simply β-equivalence. Changing this definition simplifies some things and complicates others. The definition of semantic types is simpler: whereas Chapman et al permit function types to be interpreted as `NeutralTypes`, ours must be interpreted into solely Kripke function spaces. This complicates the definitions of `reify` and `reflect`, which must become mutually recursive, as we are unable to reflect neutral types at arrow kind to neutral types. We will show later that some of Chapman et al's metatheory relies on neutral forms to not be disturbed by normalization. This complicates the definition of term-level, normality-preserving substitution.

REFERENCES

Guillaume Allais, Pierre Boutillier, and Conor McBride. New equations for neutral terms: A sound and complete decision procedure, formalized, 2013. URL https://arxiv.org/abs/1304.0809.

James Chapman, Roman Kireev, Chad Nester, and Philip Wadler. System F in agda, for fun and profit. In Graham Hutton, editor, *Mathematics of Program Construction - 13th International Conference, MPC 2019, Porto, Portugal, October 7-9, 2019, Proceedings*, volume 11825 of *Lecture Notes in Computer Science*, pages 255–297. Springer, 2019. ISBN 978-3-030-33635-6. doi: 10.1007/978-3-030-33636-3_10. URL https://doi.org/10.1007/978-3-030-33636-3_10.

Alex Hubers and J. Garrett Morris. Generic programming with extensible data types: Or, making ad hoc extensible data types less ad hoc. *Proc. ACM Program. Lang.*, 7(ICFP):356–384, 2023. doi: 10.1145/3607843. URL https://doi.org/10.1145/3607843.

Philip Wadler, Wen Kokke, and Jeremy G. Siek. *Programming Language Foundations in Agda*. August 2022. URL https://plfa.inf.ed.ac.uk/20.08/.